Usando o Maple em Geometria Analítica

Lenimar Nunes de Andrade
UFPB - João Pessoa, PB
e-mail: lenimar@mat.ufpb.br

20 de fevereiro de 2002

1 Introdução

O Maple é um programa de Computação Algébrica de uso geral que possui inúmeros recursos numéricos e gráficos, além de também funcionar como uma linguagem de programação. Ele vem sendo desenvolvido no Canadá desde 1981 pela *Waterloo Maple Inc.* Com ele, é possível realizar cálculos que contenham símbolos como π, ∞ ou $\sqrt{2}$ sem a necessidade de fazer aproximações numéricas ou realizar simplificações e cálculos com expressões algébricas como $ax^2 + bx + c$ ou $x^3 + \log(x)$ sem ser preciso atribuir valores numéricos às variáveis ou constantes. Devido a essas propriedades, é possível encontrar soluções exatas para problemas práticos que envolvam resolução de equações, derivadas, integrais, cálculo matricial, etc, tudo isso integrado a recursos que permitem visualização de gráficos planos ou tridimensionais.

O objetivo deste artigo é dar uma pequena amostra de como esse programa pode ser usado para resolver problemas de Geometria Analítica.

Cada comando digitado deve terminar com um ";" (ponto e vírgula) ou com ":" (dois pontos), seguido de Enter. Se o comando terminar com ponto e vírgula, o resultado da sua execução será mostrado logo em seguida. Se terminar com dois pontos, o resultado não será mostrado, podendo ser usado posteriormente.

As operações aritméticas básicas são indicadas por + (adição), - (subtração), * (multiplicação), / (divisão) e ^ (potenciação) e a ordem de prioridade nos cálculos é a mesma utilizada na Matemática.

Exemplo 1.1 *Ao lado do aviso de prontidão do Maple (um sinal de maior) digitamos* $3 + x^2 - 4a + 5 + 1 - 7a + x^2$ *e encerramos a linha com um ponto e vírgula. Ele calcula a soma e mostra imediatamente o resultado:*

```
> 3 + x^2 - 4*a + 5 + 1 - 7*a + x^2;
```

$$9 + 2x^2 - 11a$$

A simplificação é fundamental na apresentação de muitos resultados. Para isso, o Maple possui o comando `simplify(`*expressão*`)`, entre outros comandos.

Exemplo 1.2 *Simplificar a expressão*

$$\frac{x^6 + 3x^5 - 3x^4 - 42x^3 - 153x^2 + 3x + 11}{x^6 - 4x^5 - 15x^4 + 56x^3 + 15x^2 - 4x - 1}$$

```
> simplify( (x^6  + 3*x^5  - 3*x^4  - 42*x^3  - 153*x^2  + 3*x + 11)/
>           (x^6  - 4*x^5  - 15*x^4  + 56*x^3  + 15*x^2  - 4*x - 1 ));
```

$$\frac{x^2 + 3x + 11}{x^2 - 4x - 1}$$

Exemplo 1.3 *O comando* `factor(`*expressão*`)` *pode ser usado para fatorar uma expressão dada. Por exemplo, fatorando* $x^4 - 16$ *e* $x^5 + x + 1$ *obtemos:*

```
> factor(x^4 - 16);
```

$$(x-2)(x+2)(x^2+4)$$

```
> factor(x^5 + x + 1);
```

$$(x^2+x+1)(x^3-x^2+1)$$

Exemplo 1.4 *O comando para resolução de uma equação é o* `solve(`*equação*`)`. *Ele encontra soluções reais ou complexas de muitos tipos de equações: polinomiais, logarítmicas, exponenciais, trigonométricas, irracionais, etc. Aqui, resolvemos duas equações do segundo grau e a equação irracional* $\sqrt{x+6} + \sqrt{x+1} = \sqrt{7x+4}$. *O comando* `sqrt(x)` *denota a raiz quadrada de* x.

```
> solve( x^2 + 4*x + 2 = 0 );
```

$$-2+\sqrt{2},\ -2-\sqrt{2}$$

```
> solve( x^2 + 4*x + 5 = 0);
```

$$-2+I,\ -2-I$$

```
> solve( sqrt(x + 6) + sqrt(x + 1) = sqrt(7*x + 4));
```


2 Pontos e retas

Os comandos do Maple estão agrupados de acordo com assunto, formando pacotes que podem ser chamados com um comando `with(`*pacote*`)`. Por exemplo, o Maple possui um pacote chamado `geometry` com mais de 100 comandos relacionados com itens de geometria euclidiana plana.

No pacote `geometry` um ponto $P = (a, b)$ é definido na forma `point(P, a, b)` e uma reta L pode ser definida de duas maneiras com um comando `line`:

- Com sua equação e a lista de variáveis: `line(`*L*`,` *equação*`, [`*variáveis*`])`
- Com dois de seus pontos A, B e as variáveis: `line(`*L*`, [`*A*`,` *B*`], [`*variáveis*`])`

Entre as muitas funções do pacote `geometry` relacionados com os objetos `point` e `line`, vamos citar algumas:

coordinates(P) Coordenadas do ponto P

Equation(L) Equação da reta L

slope(L) Declividade da reta L

distance(P, Q) Distância entre os pontos P e Q

distance(P, L) Distância entre o ponto P e a reta L

PerpendicularLine(L2, P, L1) Define a reta $L2$ que passa pelo ponto P e é perpendicular à reta L1

ParallelLine(L3, Q, L4) Define a reta $L3$ que passa pelo ponto Q e é paralela à reta $L4$

Exemplo 2.1 *Obter a equação da reta que passa pelo ponto* $P = (5, -6)$ *e é perpendicular à reta que passa por* $A = (1, 1)$ *e* $B = (-3, 2)$.

Inicialmente, definimos um ponto P como objeto geométrico do pacote `geometry`.

```
> with(geometry):
> point(P, 5, -6):
```

Agora, definimos a reta r que passa pelos pontos $A = (1, 1)$ e $B = (-3, 2)$.

```
> line(r, [point(A, 1, 1), point(B, -3, 2)], [x, y]):
```

Depois de definida a reta r, podemos obter sua equação e sua declividade.

```
> Equation(r);
```

$$5 - x - 4y = 0$$

```
> slope(r);
```

$$-\frac{1}{4}$$

Definimos a reta s que passa por P e é perpendicular a r e mostramos sua equação.

```
> PerpendicularLine(s, P, r):
> Equation(s);
```

$$26 - 4x + y = 0$$

3 Circunferências

Uma circunferência C pode ser definida usando-se o comando `circle` de uma das quatro maneiras mostradas a seguir:

- Com três pontos P, Q e R e lista de variáveis: `circ(`C`, [`P, Q, R`], [`*variáveis*`])`
- Com dois de seus pontos P e Q diametralmente opostos e a lista de variáveis: `circ(`C`, [`P, Q`], [`*variáveis*`])`
- Com sua equação e a lista de variáveis: `circle(`C`, `*equação*`, [`*variáveis*`])`

- Com o centro O e o raio r: `circ(`C`, [`O, r`], [`*variáveis*`])`

Citemos algumas das funções do pacote `geometry` relacionadas com o comando `circle`:

center(C) Centro da circunferência C

radius(C) Raio da circunferência C

Equation(C) Equação da circunferência C

Exemplo 3.1 *Seja C a circunferência cuja equação é $x^2+y^2-3x+7y-1=0$. Calcular as coordenadas do seu centro e o seu raio.*

```
> with(geometry):
> circle(C, x^2 + y^2 - 3*x + 7*y - 1 = 0, [x, y]):
> coordinates(center(C));
```

$$\left[\frac{3}{2}, -\frac{7}{2}\right]$$

```
> radius(C);
```

$$\frac{1}{2}\sqrt{31}\sqrt{2}$$

O comando `solve` também pode ser usado para resolver sistemas de equações. Para isso, basta utilizá-lo com a lista entre chaves das equações do sistema.

Exemplo 3.2 *Determine a interseção da circunferência C2 que passa pelos pontos $A(5,4)$, $B(6,1)$ e $C(-3,-2)$ com a reta R1 que passa pelos pontos $D(1,2)$ e $E(-4,1)$.*

```
> with(geometry):
> circle(C2, [point(A,5,4), point(B,6,1), point(C,-3,-2)], [x,y]):
> line(R1, [point(D, 1, 2), point(E, -4, 1)], [x,y]):
> EqCirc := Equation(C2);
```

$$EqCirc := x^2 - 23 + y^2 - 2x - 2y = 0$$

O sinal de := é utilizado pelo Maple para fazer a atribuição de valor a uma determinada variável.

```
> EqReta := Equation(R1);
```

$$EqReta := 9 + x - 5y = 0$$

```
> solve({EqCirc, EqReta});
```

$$\{y = 1,\, x = -4\},\, \{y = \frac{38}{13},\, x = \frac{73}{13}\}$$

Obtivemos então uma interseção formada pelo conjunto de pontos

$$\left\{(-4,\, 1),\, (\frac{73}{13}, \frac{38}{13})\right\}$$

4 Cônicas

No pacote `geometry`, cada cônica da qual se conhece a *equação* pode ser definida com um comando

`conic(G, [`*equação*`], [`*variáveis*`])`

Opcionalmente, a palavra `conic` pode ser substituída por `parabola`, `ellipse` ou `hyperbola`.

Mas, há muitas outras maneiras de definição de cada cônica. Por exemplo, uma parábola P da qual seja conhecida o foco F e o vértice V pode ser definida com um comando

`parabola(P, [focus = F, vertex = V], [`*variáveis*`]),`

uma elipse E da qual sejam conhecidos seus focos $F1$ e $F2$ e a soma dos raios focais s pode ser definida na forma

`ellipse(E, [foci = [F1, F2], distance= s], [`*variáveis*`])`

e uma hipérbole H da qual sejam conhecidos os focos $F1$ e $F2$ e os vértices $V1$ e $V2$ pode ser definida por

`hyperbola(H, [foci = [F1, F2], vertices = [V1, V2]], [`*variáveis*`]).`

Depois de definida, podemos obter informações a respeito de uma cônica `G` com os seguintes comandos:

form(G) Classificação de G

Equation(G) Equação de G

foci(G) Focos de G

vertices(G) Vértices de G

center(G) Centro de G

asymptotes(G) Assíntotas de G

directrix(G) Diretriz de G

Exemplo 4.1 *Considere a parábola P com vértice $V = (-3, 4)$ e foco $F = (2, 4)$. Determine sua equação e sua diretriz.*

```
> with(geometry):
> parabola(P, [vertex = point(V,-3,4), focus = point(F,2,4)], [x,y]):
> EqPar := Equation(P);
```

$$EqPar := -1100 + 25y^2 - 500x - 200y = 0$$

```
> EqPar/25;
```

$$-44 + y^2 - 20x - 8y = 0$$

```
> Equation(directrix(p));
```

$$x + 8 = 0$$

Exemplo 4.2 *Determine a equação da elipse* `E` *de focos* $F_1 = (4, 3)$, $F_2 = (4, -5)$ *e soma dos raios focais igual a 12.*

```
> with(geometry):
> point(F1, 4, 3), point(F2, 4, -5):
> ellipse(E, [foci=[F1, F2], distance=12], [x, y]):
> eq := Equation(e);
```

$$eq := 576x^2 - 4608x + 320y^2 + 640y - 1984 = 0$$

Podemos usar o comando `completesquare(`*equação*`,` *variável*`)` *do pacote* `student` *para completar os quadrados e escrever a equação em um formato usual.*

```
> with(student): eq := completesquare(eq, x);
```

$$eq := 576(x-4)^2 - 11200 + 320y^2 + 640y = 0$$

```
> eq := completesquare(eq, y);
```

$$eq := 320(y+1)^2 - 11520 + 576(x-4)^2 = 0$$

Agora, somamos 11520 *ao lado esquerdo (*`lhs`*) e ao lado direito (*`rhs`*) da equação* `eq`*:*

```
> eq := lhs(eq) + 11520 = rhs(eq) + 11520;
```

$$eq := 320(y+1)^2 + 576(x-4)^2 = 11520$$

```
> eq := eq/11520;
```

$$eq := \frac{1}{36}(y+1)^2 + \frac{1}{20}(x-4)^2 = 1$$

Exemplo 4.3 *Determine os focos, os vértices e as assíntotas da hipérbole* H *de equação* $9y^2 - 4x^2 - 36y + 20x = 100$.

```
> with(geometry):
> hyperbola(H, 9*y^2 - 4*x^2 - 36*y + 20*x = 100, [x,y]):
> map(coordinates, foci(H));
```

$$[[\frac{5}{2}, 2 - \frac{1}{6}\sqrt{1443}],\ \ [\frac{5}{2}, 2 + \frac{1}{6}\sqrt{1443}]]$$

```
> map(coordinates, vertices(H));
```

$$[[\frac{5}{2}, 2 - \frac{1}{3}\sqrt{111}],\ \ [\frac{5}{2}, 2 + \frac{1}{3}\sqrt{111}]]$$

```
> map(Equation, asymptotes(H));
```

$$[y + \frac{2}{3}x - \frac{11}{3} = 0,\ \ y - \frac{2}{3}x - \frac{1}{3} = 0]$$

Exemplo 4.4 *Determine o centro e identifique a cônica definida pela equação* $2x^2 + 3y^2 + 8x - 6y - 13 = 0$.

```
> with(geometry):
> conic(c, 2*x^2 + 3*y^2  + 8*x - 6*y - 13 = 0, [x, y]):
> coordinates(center(c));
```

$$[-2,\ 1]$$

```
> form(c);
```

$$ellipse2d$$

Para obter mais informações sobre o Maple e sobre Computação Algébrica, consultar as páginas www.SymbolicNet.org e www.mapleapps.com na Internet.